<commit_before><xsl:stylesheet xmlns:xsl="http://www.w3.org/1999/XSL/Transform" version="1.0">

<xsl:output method="text" encoding="us-ascii" omit-xml-declaration="yes" indent="no"/>

 <xsl:template match="/element">
  <xsl:apply-templates select="name"/>
  <xsl:apply-templates select="details"/>
  <xsl:apply-templates select="object"/>
  <xsl:apply-templates select="pad-templates"/>
  <xsl:apply-templates select="element-flags"/>
  <xsl:apply-templates select="element-implementation"/>
  <xsl:apply-templates select="clocking-interaction"/>
  <xsl:apply-templates select="pads"/>
  <xsl:apply-templates select="element-properties"/>
  <xsl:apply-templates select="element-signals"/>
  <xsl:apply-templates select="element-actions"/>
 </xsl:template>

 <xsl:template match="name">
  <xsl:text>Element Name: </xsl:text><xsl:value-of select="."/>
  <xsl:text>&#10;&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="details">
  <xsl:text>Factory Details:&#10;</xsl:text> 
  <xsl:text>  Long Name:&#9;</xsl:text>   <xsl:value-of select="long-name"/>   <xsl:text>&#10;</xsl:text>
  <xsl:text>  Class:&#9;</xsl:text>       <xsl:value-of select="class"/>       <xsl:text>&#10;</xsl:text>
  <xsl:text>  License:&#9;</xsl:text>     <xsl:value-of select="license"/>     <xsl:text>&#10;</xsl:text>
  <xsl:text>  Description:&#9;</xsl:text> <xsl:value-of select="description"/> <xsl:text>&#10;</xsl:text>
  <xsl:text>  Version:&#9;</xsl:text>     <xsl:value-of select="version"/>     <xsl:text>&#10;</xsl:text>
  <xsl:text>  Author(s):&#9;</xsl:text>   <xsl:value-of select="authors"/>     <xsl:text>&#10;</xsl:text>
  <xsl:text>  Copyright:&#9;</xsl:text>   <xsl:value-of select="copyright"/>   <xsl:text>&#10;</xsl:text>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="object">
 </xsl:template>

 <xsl:template match="pad-templates">
  <xsl:text>Pad Templates&#10;</xsl:text>
  <xsl:apply-templates select="./pad-template"/>
 </xsl:template>

 <xsl:template match="pad-template">
  <xsl:text>  </xsl:text>
  <xsl:value-of select="direction"/> 
  <xsl:text> template: </xsl:text>
  <xsl:value-of select="name"/>
  <xsl:text>&#10;</xsl:text>
  <xsl:text>    Availability: </xsl:text> <xsl:value-of select="presence"/>
  <xsl:text>&#10;</xsl:text>
  <xsl:text>    Capabilities:&#10; </xsl:text> <xsl:apply-templates select="./capscomp"/>
 </xsl:template>

 <xsl:template match="capscomp">
  <xsl:apply-templates select="./caps"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="caps">
  <xsl:text>     '</xsl:text>
  <xsl:value-of select="name"/>
  <xsl:text>'&#10;</xsl:text>
  <xsl:text>        MIME type: </xsl:text>
  <xsl:value-of select="type"/>
  <xsl:text>'&#10;</xsl:text>
  <xsl:apply-templates select="./properties"/>
 </xsl:template>

 <xsl:template match="properties">
  <xsl:apply-templates select="*"/>
 </xsl:template>

 <xsl:template match="list">
  <xsl:text>        </xsl:text>
  <xsl:value-of select="@name"/>
  <xsl:text>&#9;:List&#10;</xsl:text>
  <xsl:apply-templates select="*" mode="list"/>
 </xsl:template>

 <!-- propety entries in list mode -->
 <xsl:template match="string" mode="list">
  <xsl:text>         String: '</xsl:text>
  <xsl:value-of select="@value"/>
  <xsl:text>'&#10;</xsl:text>
 </xsl:template>
 
 <xsl:template match="fourcc" mode="list">
  <xsl:text>         FourCC: '</xsl:text>
  <xsl:value-of select="@hexvalue"/>
  <xsl:text>'&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="int" mode="list">
  <xsl:text>         Integer: </xsl:text>
  <xsl:value-of select="@value"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="range" mode="list">
  <xsl:text>         Integer range: </xsl:text>
  <xsl:value-of select="concat(@min, ' - ', @max)"/> 
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <!-- propety entries in normal mode -->
 <xsl:template match="string">
  <xsl:text>         </xsl:text>
  <xsl:value-of select="@name"/>
  <xsl:text>     : String: '</xsl:text>
  <xsl:value-of select="@value"/> 
  <xsl:text>'&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="fourcc">
  <xsl:text>         </xsl:text>
  <xsl:value-of select="@name"/>
  <xsl:text>     : FourCC: '</xsl:text>
  <xsl:value-of select="@hexvalue"/> 
  <xsl:text>'&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="int">
  <xsl:text>         </xsl:text>
  <xsl:value-of select="@name"/>
  <xsl:text>     : Integer: </xsl:text>
  <xsl:value-of select="@value"/> 
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="range"> 		
  <xsl:text>         </xsl:text>
  <xsl:value-of select="@name"/>
  <xsl:text>     : Integer range: </xsl:text>
  <xsl:value-of select="concat(@min, ' - ', @max)"/> 
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="flag">
  <xsl:text>  </xsl:text>
  <xsl:value-of select="."/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="element-flags">
  <xsl:text>Element Flags:&#10;</xsl:text>
  <xsl:apply-templates select="./flag"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="state-change">
  <xsl:text>  Has change_state() function: </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="load">
  <xsl:text>  Has custom restore_thyself() function: </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="save">
  <xsl:text>  Has custom save_thyself() function: </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="element-implementation">
  <xsl:text>Element Implementation:&#10;</xsl:text>
  <xsl:apply-templates select="*"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="clocking-interaction">
  <xsl:text>Clocking Interaction:&#10;</xsl:text>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="pads">
  <xsl:text>Pads:&#10;</xsl:text>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="element-properties">
  <xsl:text>Element Arguments:&#10;</xsl:text>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="element-signals">
  <xsl:text>Element Signals:&#10;</xsl:text>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="element-actions">
  <xsl:text>Element Actions:&#10;</xsl:text>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

</xsl:stylesheet><commit_msg>Added pad implementation + element properties<commit_after><xsl:stylesheet xmlns:xsl="http://www.w3.org/1999/XSL/Transform" version="1.0">

<xsl:output method="text" encoding="us-ascii" omit-xml-declaration="yes" indent="no"/>

 <xsl:template match="/element">
  <xsl:apply-templates select="name"/>
  <xsl:apply-templates select="details"/>
  <xsl:apply-templates select="object"/>
  <xsl:apply-templates select="pad-templates"/>
  <xsl:apply-templates select="element-flags"/>
  <xsl:apply-templates select="element-implementation"/>
  <xsl:apply-templates select="clocking-interaction"/>
  <xsl:apply-templates select="pads"/>
  <xsl:apply-templates select="element-properties"/>
  <xsl:apply-templates select="element-signals"/>
  <xsl:apply-templates select="element-actions"/>
 </xsl:template>

 <xsl:template match="name">
  <xsl:text>Element Name: </xsl:text><xsl:value-of select="."/>
  <xsl:text>&#10;&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="details">
  <xsl:text>Factory Details:&#10;</xsl:text> 
  <xsl:text>  Long Name:&#9;</xsl:text>   <xsl:value-of select="long-name"/>   <xsl:text>&#10;</xsl:text>
  <xsl:text>  Class:&#9;</xsl:text>       <xsl:value-of select="class"/>       <xsl:text>&#10;</xsl:text>
  <xsl:text>  License:&#9;</xsl:text>     <xsl:value-of select="license"/>     <xsl:text>&#10;</xsl:text>
  <xsl:text>  Description:&#9;</xsl:text> <xsl:value-of select="description"/> <xsl:text>&#10;</xsl:text>
  <xsl:text>  Version:&#9;</xsl:text>     <xsl:value-of select="version"/>     <xsl:text>&#10;</xsl:text>
  <xsl:text>  Author(s):&#9;</xsl:text>   <xsl:value-of select="authors"/>     <xsl:text>&#10;</xsl:text>
  <xsl:text>  Copyright:&#9;</xsl:text>   <xsl:value-of select="copyright"/>   <xsl:text>&#10;</xsl:text>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="object">
 </xsl:template>

 <xsl:template match="pad-templates">
  <xsl:text>Pad Templates&#10;</xsl:text>
  <xsl:apply-templates select="./pad-template"/>
 </xsl:template>

 <xsl:template match="pad-template">
  <xsl:text>  </xsl:text>
  <xsl:value-of select="direction"/> 
  <xsl:text> template: </xsl:text>
  <xsl:value-of select="name"/>
  <xsl:text>&#10;</xsl:text>
  <xsl:text>    Availability: </xsl:text> <xsl:value-of select="presence"/>
  <xsl:text>&#10;</xsl:text>
  <xsl:text>    Capabilities:&#10; </xsl:text> <xsl:apply-templates select="./capscomp"/>
 </xsl:template>

 <xsl:template match="capscomp">
  <xsl:apply-templates select="./caps"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="caps">
  <xsl:text>     '</xsl:text>
  <xsl:value-of select="name"/>
  <xsl:text>'&#10;</xsl:text>
  <xsl:text>        MIME type: </xsl:text>
  <xsl:value-of select="type"/>
  <xsl:text>'&#10;</xsl:text>
  <xsl:apply-templates select="./properties"/>
 </xsl:template>

 <xsl:template match="properties">
  <xsl:apply-templates select="*"/>
 </xsl:template>

 <xsl:template match="list">
  <xsl:text>        </xsl:text>
  <xsl:value-of select="@name"/>
  <xsl:text>&#9;:List&#10;</xsl:text>
  <xsl:apply-templates select="*" mode="list"/>
 </xsl:template>

 <!-- propety entries in list mode -->
 <xsl:template match="string" mode="list">
  <xsl:text>         String: '</xsl:text>
  <xsl:value-of select="@value"/>
  <xsl:text>'&#10;</xsl:text>
 </xsl:template>
 
 <xsl:template match="fourcc" mode="list">
  <xsl:text>         FourCC: '</xsl:text>
  <xsl:value-of select="@hexvalue"/>
  <xsl:text>'&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="int" mode="list">
  <xsl:text>         Integer: </xsl:text>
  <xsl:value-of select="@value"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="range" mode="list">
  <xsl:text>         Integer range: </xsl:text>
  <xsl:value-of select="concat(@min, ' - ', @max)"/> 
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <!-- propety entries in normal mode -->
 <xsl:template match="string">
  <xsl:text>         </xsl:text>
  <xsl:value-of select="@name"/>
  <xsl:text>     : String: '</xsl:text>
  <xsl:value-of select="@value"/> 
  <xsl:text>'&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="fourcc">
  <xsl:text>         </xsl:text>
  <xsl:value-of select="@name"/>
  <xsl:text>     : FourCC: '</xsl:text>
  <xsl:value-of select="@hexvalue"/> 
  <xsl:text>'&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="int">
  <xsl:text>         </xsl:text>
  <xsl:value-of select="@name"/>
  <xsl:text>     : Integer: </xsl:text>
  <xsl:value-of select="@value"/> 
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="range"> 		
  <xsl:text>         </xsl:text>
  <xsl:value-of select="@name"/>
  <xsl:text>     : Integer range: </xsl:text>
  <xsl:value-of select="concat(@min, ' - ', @max)"/> 
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="flag">
  <xsl:text>  </xsl:text>
  <xsl:value-of select="."/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="element-flags">
  <xsl:text>Element Flags:&#10;</xsl:text>
  <xsl:apply-templates select="./flag"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="state-change">
  <xsl:text>  Has change_state() function: </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="load">
  <xsl:text>  Has custom restore_thyself() function: </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="save">
  <xsl:text>  Has custom save_thyself() function: </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="element-implementation">
  <xsl:text>Element Implementation:&#10;</xsl:text>
  <xsl:apply-templates select="*"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="clocking-interaction">
  <xsl:text>Clocking Interaction:&#10;</xsl:text>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="pads">
  <xsl:text>Pads:&#10;</xsl:text>
  <xsl:apply-templates select="pad"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="pad">
  <xsl:text>  </xsl:text>
  <xsl:value-of select="direction"/> 
  <xsl:text>: '</xsl:text>
  <xsl:value-of select="name"/>
  <xsl:text>'&#10;</xsl:text>
  <xsl:apply-templates select="implementation"/>
  <xsl:text>    Pad Template: '</xsl:text>
  <xsl:value-of select="template"/>
  <xsl:text>'&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="implementation">
  <xsl:text>    Implementation:&#10;</xsl:text>
  <xsl:apply-templates select="*"/>
 </xsl:template>

 <xsl:template match="chain-based">
  <xsl:text>      Has chainfunc(): </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="bufferpool-function">
  <xsl:text>      Has bufferpoolfunc(): </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="format">
  <xsl:text>                (</xsl:text>
  <xsl:value-of select="@id"/>
  <xsl:text>)&#9;</xsl:text>
  <xsl:value-of select="@nick"/>
  <xsl:text> (</xsl:text>
  <xsl:value-of select="."/>
  <xsl:text>)&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="formats-function">
  <xsl:text>      Supports seeking/conversion/query formats: </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
  <xsl:apply-templates select="format"/>
 </xsl:template>

 <xsl:template match="convert-function">
  <xsl:text>      Has custom convertfunc(): </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="query-function">
  <xsl:text>      Has custom queryfunc(): </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="event-function">
  <xsl:text>      Has custom eventfunc(): </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="event">
  <xsl:text>                </xsl:text>
  <xsl:value-of select="@type"/>
  <xsl:for-each select="flag">
   <xsl:text> | </xsl:text>
   <xsl:value-of select='.'/>
  </xsl:for-each>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="event-mask-func">
  <xsl:text>        Provides event masks: </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
  <xsl:apply-templates select="event"/>
 </xsl:template>

 <xsl:template match="query-type">
  <xsl:text>                (</xsl:text>
  <xsl:value-of select="@id"/>
  <xsl:text>)&#9;</xsl:text>
  <xsl:value-of select="@nick"/>
  <xsl:text> (</xsl:text>
  <xsl:value-of select="."/>
  <xsl:text>)&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="query-type-func">
  <xsl:text>        Provides query types: </xsl:text>
  <xsl:value-of select="@function"/>
  <xsl:text>&#10;</xsl:text>
  <xsl:apply-templates select="query-type"/>
 </xsl:template>

 <xsl:template match="element-properties">
  <xsl:text>Element Arguments:&#10;</xsl:text>
  <xsl:text>&#10;</xsl:text>
  <xsl:apply-templates select="element-property"/>
 </xsl:template>

 <xsl:template match="default">
  <xsl:text>. (Default </xsl:text>
  <xsl:value-of select="."/>
  <xsl:text>)</xsl:text>
 </xsl:template>

 <xsl:template match="element-property">
  <xsl:text>  </xsl:text>
  <xsl:value-of select="name"/>
  <xsl:text>:&#9;  </xsl:text>
  <xsl:value-of select="blurb"/>
  <xsl:text>&#10;</xsl:text>
  <xsl:text>                </xsl:text>
  <xsl:value-of select="type"/>
  <xsl:apply-templates select="default"/>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="element-signals">
  <xsl:text>Element Signals:&#10;</xsl:text>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

 <xsl:template match="element-actions">
  <xsl:text>Element Actions:&#10;</xsl:text>
  <xsl:text>&#10;</xsl:text>
 </xsl:template>

</xsl:stylesheet>
{'fraction_non_alphanumeric': 0.24012715490891307, 'fraction_numerical': 0.014977381097933732, 'mean_word_length': 4.772406492589979, 'pattern_counts': {'":': 0, '<': 666, '<?xml version=': 0, '>': 666, 'https://': 0, 'lorem ipsum': 0, 'www.': 2, 'xml': 4}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4468548', 'n_tokens_mistral': 6833, 'n_tokens_neox': 5899, 'n_words': 902}